REPRESENTATIONS FOR PARTICLES UP TO SPIN 3/2

THOMAS REITER

Abstract. This document describes the implementation of wave-functions and propagators in `GOLEM`.

Contents

1. Global Structure

The replacements for the wave-functions go into the file `legs.hh`, propagators are found in `propagators.hh`. The diagram generator is expected to yield the following functions:

`inplorentz(`$2s$`, `i`, `k`, `m`):` for each initial state particle of spin s, momentum k and mass m. The index i is a Lorentz index in the corresponding representation which connects the wave function to the rest of the diagram. In cases where particle and

Date: Last updated: September 29, 2011.

antiparticle are distinct, the parameter $2s$ is signed ($-2s$ for the antiparticle).

`outlorentz(`$2s$`, `i`, `k`, `m`):` as above, but for each final state particle.

`proplorentz(`$2s$`, `k`, `m`, `Γ`, `A`, `i_1`, `i_2`):` denotes the Lorentz part of a propagator for a particle of spin s, momentum k and mass m. The indices i_1 and i_2 are corresponding Lorentz indices. The decay width of the particle is Γ. The parameter A is a flag that indicates special properties of a field and is non-zero if the propagator needs special treatment.

`inpcolor(`n`, `i`):` for each initial state particle. Associates the colour index i with the initial state particle number n. This function is not treated in this file.

`outcolor(`n`, `i`):` for each final state particle. Associates the colour index i with the finaltial state particle number n. This function is not treated in this file.

`propcolor(`r`, `i_1`, `i_2`):` denotes the colour part of a propagator, where r is a representation label; r is either 1 (trivial rep.), -3 or 3 (fundamental rep.) or 8 (adjoint rep.). The indices i_1 and i_2 are the colour indices of that propagator.

`inp(`f`, `k`, [`h`], [`$k^\flat$`], [`q`]):` carries the helicity information h of an initial state particle of the field f and momentum k. For massive gauge bosons, the parameters $k^\flat$ and q are the two momenta of the light-cone splitting. For massless gauge the parameter $k^\flat$ is ommitted. The parameters h, $k^\flat$ and q are not generated by the diagram generator but added at an earlier point in the `Form` program.

`out(`f`, `k`, [`h`], [`$k^\flat$`], [`q`]):` same as `inp` but for final state particles.

On the output side we use the symbols introduced by the `spinney` library plus the scalar propagator

(1)
$$\texttt{inv}(k,m)=\frac{1}{k^2-m^2+i0^+}\quad\text{and}\quad\texttt{inv}(k,m,\Gamma)=\frac{1}{k^2-m^2-im\Gamma+i0^+}$$

⟨ *common header* ? ⟩ ≡

```
* vim: ts=3:sw=3
```

◇

Macro referenced in ?, ?.

`"legs.hh"` ?≡

⟨ *common header* ? ⟩

```
*---#[ Scalars :
```

⟨ *scalar wave-functions* ? ⟩

```
*---#] Scalars :
*---#[ Spinors :
```

```
*---#[   Massless Spinors :
⟨ wave-functions for massless spinors ? ⟩
*---#]   Massless Spinors :
*---#[   Massive Spinors :
⟨ wave-functions for massive spinors ? ⟩
*---#]   Massive Spinors :
*---#] Spinors :
*---#[ Polarisation Vectors for Gauge Bosons :
*---#[    Massless Gauge Bosons :
⟨ gauge boson wave-functions, light-like ? ⟩
*---#]    Massless Gauge Bosons :
*---#[    Massive Gauge Bosons :
⟨ gauge boson wave-functions, massive ? ⟩
*---#]    Massive Gauge Bosons :
*---#] Polarisation Vectors for Gauge Bosons :
*---#[ wave functions for Vector-Spinors :
Repeat;
   ⟨ vector-spinor wave functions ? ⟩
EndRepeat;
*---#] wave functions for Vector-Spinors :
*---#[ wave functions for gravitons :
Repeat;
   ⟨ graviton wave functions ? ⟩
EndRepeat;
*---#] wave functions for gravitons :
◇
```

```
"propagators.hh" ?≡
      ⟨ common header ? ⟩
        ⟨ colour part of the propagators ? ⟩
        *---#[ Scalar Bosons :
        ⟨ scalar propagator ? ⟩
        *---#] Scalar Bosons :
        *---#[ Fermions :
        ⟨ fermion propagator ? ⟩
        ⟨ handed fermion propagator ? ⟩
        *---#] Fermions :
        *---#[ Gauge Bosons :
        ⟨ gauge boson propagator ? ⟩
        *---#] Gauge Bosons :
        *---#[ Vector-Spinor propagator :
        Repeat;
           ⟨ vector-spinor propagator ? ⟩
        EndRepeat;
        *---#] Vector-Spinor propagator :
        *---#[ Tensor Bosons :
        Repeat;
```

⟨ *tensor ghost propagator* ? ⟩
⟨ *graviton propagator* ? ⟩
EndRepeat;
```
*---#] Tensor Bosons :
```
◇

For the Feynman rules we stick to the conventions of [1].

2. Spin-0 Particles

The wave function of a spin-0 particle is represented by a pure number.

⟨ *scalar wave-functions* ? ⟩ ≡
```
Id inplorentz(0, iv?, k1?, m?) = 1;
  Id outlorentz(0, iv?, k1?, m?) = 1;◇
```
Macro referenced in ?.

Its propagator is just the plain propagator

$$\frac{i}{k^2 - m^2 - im\Gamma + i0^+}. \tag{2}$$

⟨ *scalar propagator* ? ⟩ ≡
```
Id proplorentz(0, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
     PREFACTOR(i_) * inv(k1, m, sDUMMY1);
  Id proplorentz(0, 0, m?, sDUMMY1?, 0,iv1?, iv2?) =
     PREFACTOR(i_) * inv(ZERO, m, sDUMMY1);◇
```
Macro referenced in ?.

3. Spin-1/2 Particles

For spinor wave functions we have the following assignment in the notation of [1]:

	l^-, q	l^+, $\bar{q}$
initial	$u_\alpha(k, j_3)$	$\bar{v}_\alpha(k, j_3)$
final	$\bar{u}_\alpha(k, j_3)$	$v_\alpha(k, j_3)$

Here, l and q stand for leptons and quarks respectively. The index α denotes a spinor index and j_3 is the 3-component of the spin. We label the states by $j_3 = \pm 1$ instead of the physical values $j_3 = \pm 1/2$.

The propagator both for the massive and the massless case is

$$\frac{i(\not{k} + m)_{\alpha\beta}}{k^2 - m^2 - im\Gamma + i0^+} \tag{3}$$

where the momentum flow is from β to α.

⟨ *fermion propagator* ? ⟩ ≡

```
      Id proplorentz(1, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
         PREFACTOR(i_) * (NCContainer(Sm(k1), iv2, iv1)
          + m * NCContainer(1, iv2, iv1)
         ) * inv(k1, m, sDUMMY1);
        Id proplorentz(1, 0, m?, sDUMMY1?, 0, iv1?, iv2?) =
          + PREFACTOR(i_ * m) * NCContainer(1, iv2, iv1) * inv(ZERO, m, sDUMMY1);◇
```

Macro referenced in ?.

For massless fermions, the auxilliary field can also have values 1 and −1 for left- and right-handed particles. This follows the CalcHEP convention.

Golem	CalcHEP	Expression
+1	'L'	$\frac{\not{p}\Pi_+}{p^2}$
−1	'R'	$\frac{\not{p}\Pi_-}{p^2}$

⟨ *handed fermion propagator* ? ⟩ ≡

```
      Id proplorentz(1, k1?, 0, 0, 1, iv1?, iv2?) =
         PREFACTOR(i_) * NCContainer(Sm(k1)*ProjPlus, iv2, iv1) * inv(k1, 0);
        Id proplorentz(1, k1?, 0, 0, -1, iv1?, iv2?) =
         PREFACTOR(i_) * NCContainer(Sm(k1)*ProjMinus, iv2, iv1) * inv(k1, 0);◇
```

Macro referenced in ?.

3.1. **Massless Case.** For massless spinors we translate the spin states directly into helicity eigenstates as follows[1]:

$$u_\alpha(k,+1) = |k\rangle \qquad \bar{u}_\alpha(k,+1) = [k| \tag{4a}$$

$$u_\alpha(k,-1) = |k] \qquad \bar{u}_\alpha(k,-1) = \langle k| \tag{4b}$$

$$v_\alpha(k,+1) = |k] \qquad \bar{v}_\alpha(k,+1) = \langle k| \tag{4c}$$

$$v_\alpha(k,-1) = |k\rangle \qquad \bar{v}_\alpha(k,-1) = [k| \tag{4d}$$

⟨ *wave-functions for massless spinors* ? ⟩ ≡

⟨ *implementation of Equation* (4a) ? ⟩
⟨ *implementation of Equation* (4b) ? ⟩
⟨ *implementation of Equation* (4c) ? ⟩
⟨ *implementation of Equation* (4d) ? ⟩◇

Macro referenced in ?.

[1]Please, refer to the `spinney` documentation for notational conventions of bra- and ket-spinors.

⟨ *implementation of Equation* (4a) ? ⟩ ≡
```
        Id inplorentz( 1, iv?, k1?, 0) *
                inp(field1?, k1?,  1) =
             NCContainer(USpa(k1), iv);
           Id outlorentz( 1, iv?, k1?, 0) *
                out(field1?, k1?,  1) =
             NCContainer(UbarSpb(k1), iv);◇
```
Macro referenced in ?.

⟨ *implementation of Equation* (4b) ? ⟩ ≡
```
        Id inplorentz( 1, iv?, k1?, 0) *
                inp(field1?, k1?, -1) =
             NCContainer(USpb(k1), iv);
           Id outlorentz( 1, iv?, k1?, 0) *
                out(field1?, k1?, -1) =
             NCContainer(UbarSpa(k1), iv);◇
```
Macro referenced in ?.

⟨ *implementation of Equation* (4c) ? ⟩ ≡
```
        Id outlorentz(-1, iv?, k1?, 0) *
                out(field1?, k1?,  1) =
             NCContainer(USpb(k1), iv);
           Id inplorentz(-1, iv?, k1?, 0) *
                inp(field1?, k1?,  1) =
             NCContainer(UbarSpa(k1), iv);◇
```
Macro referenced in ?.

⟨ *implementation of Equation* (4d) ? ⟩ ≡
```
        Id outlorentz(-1, iv?, k1?, 0) *
                out(field1?, k1?, -1) =
             NCContainer(USpa(k1), iv);
           Id inplorentz(-1, iv?, k1?, 0) *
                inp(field1?, k1?, -1) =
             NCContainer(UbarSpb(k1), iv);◇
```
Macro referenced in ?.

3.2. **Massive Case.** Massive spinors translate to spinney notation in the following sense:

(5a) $u_\alpha(k,+1) = |k^+\rangle \qquad \bar{u}_\alpha(k,+1) = [k^+|$

(5b) $u_\alpha(k,-1) = |k^+] \qquad \bar{u}_\alpha(k,-1) = \langle k^+|$

(5c) $v_\alpha(k,+1) = |k^-] \qquad \bar{v}_\alpha(k,+1) = \langle k^-|$

(5d) $v_\alpha(k,-1) = |k^-\rangle \qquad \bar{v}_\alpha(k,-1) = [k^-|$

⟨ *wave-functions for massive spinors ?* ⟩ ≡
⟨ *implementation of Equation* (5a) ? ⟩
⟨ *implementation of Equation* (5b) ? ⟩
⟨ *implementation of Equation* (5c) ? ⟩
⟨ *implementation of Equation* (5d) ? ⟩⋄
Macro referenced in ?.

⟨ *implementation of Equation* (5a) ? ⟩ ≡
```
Id inplorentz( 1, iv?, k1?, m?) *
      inp(field1?, k1?,  1) =
    NCContainer(USpa(k1, +1), iv);
 Id outlorentz( 1, iv?, k1?, m?) *
      out(field1?, k1?,  1) =
    NCContainer(UbarSpb(k1, +1), iv);⋄
```
Macro referenced in ?.

⟨ *implementation of Equation* (5b) ? ⟩ ≡
```
Id inplorentz( 1, iv?, k1?, m?) *
      inp(field1?, k1?, -1) =
    NCContainer(USpb(k1, +1), iv);
 Id outlorentz( 1, iv?, k1?, m?) *
      out(field1?, k1?, -1) =
    NCContainer(UbarSpa(k1, +1), iv);⋄
```
Macro referenced in ?.

⟨ *implementation of Equation* (5c) ? ⟩ ≡
```
Id outlorentz(-1, iv?, k1?, m?) *
      out(field1?, k1?,  1) =
    NCContainer(USpb(k1, -1), iv);
 Id inplorentz(-1, iv?, k1?, m?) *
      inp(field1?, k1?,  1) =
    NCContainer(UbarSpa(k1, -1), iv);⋄
```
Macro referenced in ?.

⟨ *implementation of Equation* (5d) ? ⟩ ≡
```
Id outlorentz(-1, iv?, k1?, m?) *
      out(field1?, k1?, -1) =
    NCContainer(USpa(k1, -1), iv);
 Id inplorentz(-1, iv?, k1?, m?) *
      inp(field1?, k1?, -1) =
    NCContainer(UbarSpb(k1, -1), iv);⋄
```
Macro referenced in ?.

4. Spin-1 Particles

For ingoing gauge bosons we use the polarisation vector $\varepsilon_\mu(k, j_3)$, and for outgoing particles its conjugate $\varepsilon^*_\mu(k, j_3)$ in accordance with the notation of [1]. For internal particles we work in Feynman gauge and hence get the propagator

$$\frac{-ig^{\mu\nu}}{k^2 - m^2 - im\Gamma + i0^+}. \tag{6}$$

⟨ *gauge boson propagator* ? ⟩ ≡

```
Id proplorentz(2, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
      - PREFACTOR(i_) * d(iv1, iv2) * inv(k1, m, sDUMMY1);
   Id proplorentz(2, 0, m?, sDUMMY1?, 0, iv1?, iv2?) =
      - PREFACTOR(i_) * d(iv1, iv2) * inv(ZERO, m, sDUMMY1);◇
```

Macro referenced in ?.

4.1. Massless Case.

We represent massless gauge bosons in the way proposed by [2],

$$\varepsilon_\mu(k, +1) = \frac{\langle q|\gamma_\mu|k]}{\sqrt{2}\langle qk\rangle} \tag{7}$$

$$\varepsilon_\mu(k, -1) = \frac{[q|\gamma_\mu|k\rangle}{\sqrt{2}[kq]} \tag{8}$$

which requires an arbitrary, light-like auxilliary vector q. It follows that in this representation

$$(\varepsilon_\mu(k, \pm 1))^* = \varepsilon_\mu(k, \mp 1). \tag{9}$$

Below we implement the above expressions with the notation $k =$ `k1`, $q =$ `vDUMMY1` and $\mu =$ `ivL2`.

⟨ *expression for* $\varepsilon(k, +1)$? ⟩ ≡

```
1/sqrt2 * SpDenominator(Spa2(vDUMMY1, k1)) *
          UbarSpa(vDUMMY1) * Sm(ivL2) * USpb(k1)◇
```

Macro referenced in ?, ?.

⟨ *expression for* $\varepsilon(k, -1)$? ⟩ ≡

```
1/sqrt2 * SpDenominator(Spb2(k1, vDUMMY1)) *
          UbarSpb(vDUMMY1) * Sm(ivL2) * USpa(k1)◇
```

Macro referenced in ?, ?.

Using Equation (9) we can also define macros for the conjugate vectors

⟨ *expression for* $\varepsilon^*(k, +1)$? ⟩ ≡

⟨ *expression for* $\varepsilon(k, -1)$? ⟩◇

Macro referenced in ?.

⟨ *expression for* $\varepsilon^*(k,-1)$? ⟩ ≡
⟨ *expression for* $\varepsilon(k,+1)$? ⟩◇
Macro referenced in ?.

The usual properties for polarisation vectors are easy to prove. The polarisation vector is transverse both to k and q:

$$\varepsilon(k,j_3)\cdot k = 0, \tag{10}$$
$$\varepsilon(k,j_3)\cdot q = 0. \tag{11}$$

The polarisation vectors fulfill the completeness relation of an axial gauge,

$$\sum_{j_3=\pm1}\varepsilon^\mu(k,j_3)\left(\varepsilon^\nu(k,j_3)\right)^* = -g^{\mu\nu}+\frac{k^\mu q^\nu + k^\nu q^\mu}{k\cdot q}. \tag{12}$$

By making use of the Schouten identity one can show that a change of the auxilliary vector $q\to p$ amounts to a term proportional to k_μ,

$$\frac{\langle q|\gamma_\mu|k]}{\sqrt{2}\langle qk\rangle} = \frac{\langle p|\gamma_\mu|k]}{\sqrt{2}\langle pk\rangle} + \frac{\sqrt{2}\langle pq\rangle}{\langle pk\rangle\langle qk\rangle}k_\mu \tag{13a}$$
$$\frac{[q|\gamma_\mu|k\rangle}{\sqrt{2}[kq]} = \frac{[p|\gamma_\mu|k\rangle}{\sqrt{2}[kp]} + \frac{\sqrt{2}[qp]}{[kp][kq]}k_\mu \tag{13b}$$

⟨ *gauge boson wave-functions, light-like* ? ⟩ ≡
```
Id outlorentz(2, ivL2?, k1?, 0) *
        out(field1?, k1?,  1, vDUMMY1?) =
    ⟨ expression for ε*(k,+1) ? ⟩;
  Id outlorentz(2, ivL2?, k1?, 0) *
        out(field1?, k1?, -1, vDUMMY1?) =
    ⟨ expression for ε*(k,−1) ? ⟩;
  Id inplorentz(2, ivL2?, k1?, 0) *
        inp(field1?, k1?,  1, vDUMMY1?) =
    ⟨ expression for ε(k,+1) ? ⟩;
  Id inplorentz(2, ivL2?, k1?, 0) *
        inp(field1?, k1?, -1, vDUMMY1?) =
    ⟨ expression for ε(k,−1) ? ⟩;◇
```
Macro referenced in ?.

4.2. **Massive Case.** For the polarisation vectors of massive gauge bosons, where $k^2 = m^2$, we require[2]

$$\varepsilon(k, j_3) \cdot k = 0 \qquad \text{transverality,} \tag{14}$$

$$\varepsilon(k, j_3) \cdot \varepsilon(k, j_3') = -\delta_{j_3 j_3'} \qquad \text{orthonormality and} \tag{15}$$

$$\sum_{j_3=-1}^{1} \varepsilon_\mu(k, j_3) \left(\varepsilon_\nu(k, j_3)\right)^* = -g^{\mu\nu} + \frac{k^\mu k^\nu}{m^2} \qquad \text{completeness.} \tag{16}$$

We choose a representation based on the splitting of the momentum k into a pair of light-like vector $k^\flat$ and q, as it is implemented in the `spinney` procedure `SpLightConeDecomposition`.

$$k = k^\flat + \frac{m^2}{2k^\flat \cdot q} q \tag{17}$$

Similarly to the massless case, two of the polarisations can be chosen as

$$\varepsilon_\mu(k, +1) = \frac{\langle q|\gamma_\mu|k^\flat]}{\sqrt{2}\langle q k^\flat\rangle} \quad \text{and} \tag{18}$$

$$\varepsilon_\mu(k, -1) = \frac{[q|\gamma_\mu|k^\flat\rangle}{\sqrt{2}[k^\flat q]}. \tag{19}$$

As before, these vectors have the property that they are complex conjugate to each other. The third polarisation vector is

$$\varepsilon_\mu(k, 0) = \frac{1}{m}\left(k^\flat_\mu - \frac{m^2}{2k^\flat \cdot q} q_\mu\right) = \frac{1}{m}\left(2k^\flat_\nu - k_\nu\right). \tag{20}$$

In the implementation we have $\mu =$ `ivL2`, $k =$ `k1`, $k^\flat =$ `k2` and $q =$ `k3`.

⟨ *expression for massive* $\varepsilon(k, +1)$? ⟩ ≡

```
(1/sqrt2 * SpDenominator(Spa2(k3, k2))) *
     UbarSpa(k3) * Sm(ivL2) * USpb(k2)◇
```

Macro referenced in ?, ?, ?, ?, ?.

⟨ *expression for massive* $\varepsilon(k, -1)$? ⟩ ≡

```
(1/sqrt2 * SpDenominator(Spb2(k2, k3))) *
     UbarSpb(k3) * Sm(ivL2) * USpa(k2)◇
```

Macro referenced in ?, ?, ?, ?, ?.

⟨ *expression for massive* $\varepsilon(k, 0)$? ⟩ ≡

```
(1/m) * (k2(ivL2) - m * SpDenominator(Spa2(k2,k3)) *
     m * SpDenominator(Spb2(k3,k2)) * k3(ivL2))◇
```

Macro referenced in ?, ?, ?, ?, ?.

[2] See for example Appendix A.1.1.6 of [1].

The conjugate polarisation vectors are as follows.

⟨ *expression for massive* $\varepsilon^*(k,+1)$? ⟩ ≡
⟨ *expression for massive* $\varepsilon(k,-1)$? ⟩⋄
Macro referenced in ?, ?, ?, ?.

⟨ *expression for massive* $\varepsilon^*(k,-1)$? ⟩ ≡
⟨ *expression for massive* $\varepsilon(k,+1)$? ⟩⋄
Macro referenced in ?, ?, ?, ?.

⟨ *expression for massive* $\varepsilon^*(k,0)$? ⟩ ≡
⟨ *expression for massive* $\varepsilon(k,0)$? ⟩⋄
Macro referenced in ?, ?, ?, ?.

Finally, we can express all six possibilities of initial state and final state polarisation vectors:

⟨ *gauge boson wave-functions, massive* ? ⟩ ≡
```
Id outlorentz(2, ivL2?, k1?, m?) *
       out(field1?, k1?,  1, k2?, k3?) =
     ⟨ expression for massive ε*(k,+1) ? ⟩;
  Id outlorentz(2, ivL2?, k1?, m?) *
       out(field1?, k1?, -1, k2?, k3?) =
     ⟨ expression for massive ε*(k,−1) ? ⟩;
  Id outlorentz(2, ivL2?, k1?, m?) *
       out(field1?, k1?,  0, k2?, k3?) =
     ⟨ expression for massive ε*(k,0) ? ⟩;
  Id inplorentz(2, ivL2?, k1?, m?) *
       inp(field1?, k1?,  1, k2?, k3?) =
     ⟨ expression for massive ε(k,+1) ? ⟩;
  Id inplorentz(2, ivL2?, k1?, m?) *
       inp(field1?, k1?, -1, k2?, k3?) =
     ⟨ expression for massive ε(k,−1) ? ⟩;
  Id inplorentz(2, ivL2?, k1?, m?) *
       inp(field1?, k1?,  0, k2?, k3?) =
     ⟨ expression for massive ε(k,0) ? ⟩;⋄
```
Macro referenced in ?.

TODO:

In cases where no massless vectors are in the process ***GOLEM*** *chooses the procedure* ***SpLightConeSplitting*** *where a pair of massive vectors P, Q is split into a pair of light-like vectors p, q. The corresponding formulæ for polarisation vectors have to be worked out. Since this case is for very specific processes only we leave this for the future.*

5. Spin-$\frac{3}{2}$ Particles

For the implementation of massive Spin-$\frac{3}{2}$ fields we follow [3].
The projector is

(21)
$$\Pi^{\mu\nu} = (\not{p} + m)\left(\frac{p^\mu p^\nu}{m^2} - g^{\mu\nu}\right) - \frac{1}{3}\left(\gamma^\mu + \frac{k^\mu}{m}\right)(\not{k} - m)\left(\gamma^\nu + \frac{k^\nu}{m}\right).$$

⟨ *vector-spinor propagator ?* ⟩ ≡

```
    Id once proplorentz(3, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
         PREFACTOR(i_) *
         SplitLorentzIndex(iv1, iv1L2, iv1L1) *
         SplitLorentzIndex(iv2, iv2L2, iv2L1) * 1/3 * (
            + 4*k1(iv1L2)*k1(iv2L2)/m
            - 3*d(iv1L2,iv2L2)*m
            + 2*NCContainer(Sm(k1),iv1L1,iv2L1)*k1(iv1L2)*k1(iv2L2)/m^2
            - 3*NCContainer(Sm(k1),iv1L1,iv2L1)*d(iv1L2,iv2L2)
            - NCContainer(Sm(k1)*Sm(iv2L2),iv1L1,iv2L1)*k1(iv1L2)/m
            + NCContainer(Sm(iv1L2),iv1L1,iv2L1)*k1(iv2L2)
            - NCContainer(Sm(iv1L2)*Sm(k1),iv1L1,iv2L1)*k1(iv2L2)/m
            - NCContainer(Sm(iv1L2)*Sm(k1)*Sm(iv2L2),iv1L1,iv2L1)
            + NCContainer(Sm(iv1L2)*Sm(iv2L2),iv1L1,iv2L1)*m
            + NCContainer(Sm(iv2L2),iv1L1,iv2L1)*k1(iv1L2)
         ) * inv(k1, m, sDUMMY1);
      Sum iv1L2, iv1L1, iv2L2, iv2L1;◇
```

Macro referenced in ?.

A set of eigenvectors is provided by the following five states:

(22)
$$\epsilon^\mu_{+3/2}(p) = \epsilon^\mu_+(p)\epsilon_+(p)$$

(23)
$$\epsilon^\mu_{+1/2}(p) = \frac{1}{\sqrt{3}}\epsilon^\mu_+(p)\epsilon_-(p) + \sqrt{\frac{2}{3}}\epsilon^\mu_0(p)\epsilon_+(p)$$

(24)
$$\epsilon^\mu_{-1/2}(p) = \frac{1}{\sqrt{3}}\epsilon^\mu_-(p)\epsilon_+(p) + \sqrt{\frac{2}{3}}\epsilon^\mu_0(p)\epsilon_-(p)$$

(25)
$$\epsilon^\mu_{-3/2}(p) = \epsilon^\mu_-(p)\epsilon_-(p)$$

There are sixteen different cases

- in-/outgoing
- particle/anti-particle
- polarisation $-\frac{3}{2}, -\frac{1}{2}, \frac{1}{2}, \frac{3}{2}$

⟨ *vector-spinor wave functions ?* ⟩ ≡
⟨ *ingoing vector-spinor particle ?* ⟩
⟨ *ingoing vector-spinor anti-particle ?* ⟩
⟨ *outgoing vector-spinor particle ?* ⟩
⟨ *outgoing vector-spinor anti-particle ?* ⟩◇
Macro referenced in ?.

⟨ *ingoing vector-spinor particle ?* ⟩ ≡

```
Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
        inp(field1?, k1?, -2, k2?, k3?) =
    SplitLorentzIndex(ivL, ivL2, ivL1) *
    ⟨ expression for massive ε(k,−1) ? ⟩ *
    NCContainer(USpb(k1,+1), ivL1);
    Sum ivL2, ivL1;
  Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
        inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
    SplitLorentzIndex(ivL, ivL2, ivL1) * (
    + ⟨ expression for massive ε(k,−1) ? ⟩ *
      NCContainer(USpa(k1,+1), ivL1)
    + sqrt2 * ⟨ expression for massive ε(k,0) ? ⟩ *
      NCContainer(USpb(k1,+1), ivL1));
    Sum ivL2, ivL1;
  Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
        inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
    SplitLorentzIndex(ivL, ivL2, ivL1) * (
    + ⟨ expression for massive ε(k,+1) ? ⟩ *
      NCContainer(USpb(k1,+1), ivL1)
    + sqrt2 * ⟨ expression for massive ε(k,0) ? ⟩ *
      NCContainer(USpa(k1,+1), ivL1));
    Sum ivL2, ivL1;
  Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
        inp(field1?, k1?, +2, k2?, k3?) =
    SplitLorentzIndex(ivL, ivL2, ivL1) *
    ⟨ expression for massive ε(k,+1) ? ⟩ *
    NCContainer(USpa(k1,+1), ivL1);
    Sum ivL2, ivL1;
```

◇
Macro referenced in ?.

⟨ *ingoing vector-spinor anti-particle ?* ⟩ ≡

```
      Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, -2, k2?, k3?) =
         SplitLorentzIndex(ivL, ivL2, ivL1) *
         ⟨ expression for massive ε(k,−1) ? ⟩ *
         NCContainer(UbarSpb(k1,-1), ivL1);
         Sum ivL2, ivL1;
       Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
         SplitLorentzIndex(ivL, ivL2, ivL1) * (
         + ⟨ expression for massive ε(k,−1) ? ⟩ *
           NCContainer(UbarSpa(k1,-1), ivL1)
         + sqrt2 * ⟨ expression for massive ε(k,0) ? ⟩ *
           NCContainer(UbarSpb(k1,-1), ivL1));
         Sum ivL2, ivL1;
       Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
         SplitLorentzIndex(ivL, ivL2, ivL1) * (
         + ⟨ expression for massive ε(k,+1) ? ⟩ *
           NCContainer(UbarSpb(k1,-1), ivL1)
         + sqrt2 * ⟨ expression for massive ε(k,0) ? ⟩ *
           NCContainer(UbarSpa(k1,-1), ivL1));
         Sum ivL2, ivL1;
       Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, +2, k2?, k3?) =
         SplitLorentzIndex(ivL, ivL2, ivL1) *
         ⟨ expression for massive ε(k,+1) ? ⟩ *
         NCContainer(UbarSpa(k1,-1), ivL1);
         Sum ivL2, ivL1;
```

◇

Macro referenced in ?.

⟨ *outgoing vector-spinor particle ?* ⟩ ≡

```
Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
         inp(field1?, k1?, -2, k2?, k3?) =
      SplitLorentzIndex(ivL, ivL2, ivL1) *
      ⟨ expression for massive ε*(k,−1) ? ⟩ *
      NCContainer(UbarSpa(k1,+1), ivL1);
      Sum ivL2, ivL1;
   Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
         inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
      SplitLorentzIndex(ivL, ivL2, ivL1) * (
      + ⟨ expression for massive ε*(k,−1) ? ⟩ *
        NCContainer(UbarSpb(k1,+1), ivL1)
      + sqrt2 * ⟨ expression for massive ε*(k,0) ? ⟩ *
        NCContainer(UbarSpa(k1,+1), ivL1));
      Sum ivL2, ivL1;
   Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
         inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
      SplitLorentzIndex(ivL, ivL2, ivL1) * (
      + ⟨ expression for massive ε*(k,+1) ? ⟩ *
        NCContainer(UbarSpa(k1,+1), ivL1)
      + sqrt2 * ⟨ expression for massive ε*(k,0) ? ⟩ *
        NCContainer(UbarSpb(k1,+1), ivL1));
      Sum ivL2, ivL1;
   Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
         inp(field1?, k1?, +2, k2?, k3?) =
      SplitLorentzIndex(ivL, ivL2, ivL1) *
      ⟨ expression for massive ε*(k,+1) ? ⟩ *
      NCContainer(UbarSpb(k1,+1), ivL1);
      Sum ivL2, ivL1;
```

◇

Macro referenced in ?.

⟨ *outgoing vector-spinor anti-particle ?* ⟩ ≡

```
      Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, -2, k2?, k3?) =
         SplitLorentzIndex(ivL, ivL2, ivL1) *
         ⟨ expression for massive ε*(k,−1) ? ⟩ *
         NCContainer(USpa(k1,-1), ivL1);
         Sum ivL2, ivL1;
      Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
         SplitLorentzIndex(ivL, ivL2, ivL1) * (
         + ⟨ expression for massive ε*(k,−1) ? ⟩ *
           NCContainer(USpb(k1,-1), ivL1)
         + sqrt2 * ⟨ expression for massive ε*(k,0) ? ⟩ *
           NCContainer(USpa(k1,-1), ivL1));
         Sum ivL2, ivL1;
      Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
         SplitLorentzIndex(ivL, ivL2, ivL1) * (
         + ⟨ expression for massive ε*(k,+1) ? ⟩ *
           NCContainer(USpa(k1,-1), ivL1)
         + sqrt2 * ⟨ expression for massive ε*(k,0) ? ⟩ *
           NCContainer(USpb(k1,-1), ivL1));
         Sum ivL2, ivL1;
      Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
            inp(field1?, k1?, +2, k2?, k3?) =
         SplitLorentzIndex(ivL, ivL2, ivL1) *
         ⟨ expression for massive ε*(k,+1) ? ⟩ *
         NCContainer(USpb(k1,-1), ivL1);
         Sum ivL2, ivL1;
```

◇

Macro referenced in ?.

6. Spin-2 Particles

6.1. **Tensor Structure.** In order to map the pair of Lorentz indices into a single Multi-Index we use the function `SplitLorentzIndex`; the first argument denotes the multi-index, the second and the last argument are the two Lorentz indices.

6.2. **Tensor Ghost.** The CalcHEP way of treating colour requires the introduction of the so-called *tensor ghost*. This auxilliary field is introduced in order to split the four-gluon vertex into a pair of gluon-gluon-ghost vertices. The propagator therefore is not dynamical and has the form

$$P(T^{\mu_1\nu_1}(p_1), T^{\mu_2\nu_2}(p_2)) = -ig^{\mu_1\mu_2}g^{\nu_1\nu_2} \tag{26}$$

Tensor ghosts are indicated by having an auxilliary field value of 1.

⟨ *tensor ghost propagator* ? ⟩ ≡

```
Id once proplorentz(4, k1?, m?, sDUMMY1?, 1, iv1?, iv2?) =
      - PREFACTOR(i_) *
        SplitLorentzIndex(iv1, iv1a, iv1b) *
        SplitLorentzIndex(iv2, iv2a, iv2b) *
        d(iv1a, iv2a) * d(iv1b, iv2b);
  Sum iv1a, iv1b, iv2a, iv2b;◇
```

Macro referenced in ?.

6.3. Gravitons.

$$\begin{aligned}
\Pi^{\mu\nu,\alpha\beta}(p) &= \frac{1}{2}\left(g^{\mu\alpha}g^{\nu\beta}+g^{\nu\alpha}g^{\mu\beta}-g^{\mu\nu}g^{\alpha\beta}\right) \\
&- \frac{1}{2m^2}\left(g^{\mu\alpha}p^{\nu}p^{\beta}+g^{\nu\alpha}p^{\mu}p^{\beta}+g^{\mu\beta}p^{\nu}p^{\alpha}+g^{\nu\beta}p^{\mu}p^{\alpha}\right) \\
&+ \frac{1}{6}\left(g^{\mu\nu}+2\frac{p^{\mu}p^{\nu}}{m^2}\right)\left(g^{\alpha\beta}+2\frac{p^{\alpha}p^{\beta}}{m^2}\right)
\end{aligned} \tag{27}$$

⟨ *graviton propagator* ? ⟩ ≡

```
Id once proplorentz(4, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
     SplitLorentzIndex(iv1, iv1a, iv1b) *
     SplitLorentzIndex(iv2, iv2a, iv2b) *
     (
        + 1/2 * (
           + d(iv1a, iv2a) * d(iv1b, iv2b)
           + d(iv1b, iv2a) * d(iv1a, iv2b)
           - d(iv1a, iv1b) * d(iv2a, iv2b)
        )
        - 1/2/m^2 * (
           + d(iv1a, iv2a) * k1(iv1b) * k1(iv2b)
           + d(iv1b, iv2a) * k1(iv1a) * k1(iv2b)
           + d(iv1a, iv2b) * k1(iv1b) * k1(iv2a)
           + d(iv1b, iv2b) * k1(iv1a) * k1(iv2a)
        )
        + 1/6 * (d(iv1a, iv1b) + 2*k1(iv1a)*k1(iv1b)/m^2) *
                (d(iv2a, iv2b) + 2*k1(iv2a)*k1(iv2b)/m^2)
     ) * inv(k1, m, sDUMMY1);
  ◇
```

Macro referenced in ?.

$$\epsilon^{\mu\nu}_{\pm 2}(p) = \epsilon^{\mu}_{\pm}(p)\epsilon^{\nu}_{\pm}(p) \tag{28}$$

$$\epsilon^{\mu\nu}_{\pm 1}(p) = \frac{1}{\sqrt{2}}\left(\epsilon^{\mu}_{\pm}(p)\epsilon^{\nu}_{0}(p) + \epsilon^{\nu}_{\pm}(p)\epsilon^{\mu}_{0}(p)\right) \tag{29}$$

$$\epsilon^{\mu\nu}_{0}(p) = \frac{1}{\sqrt{6}}\left(\epsilon^{\mu}_{+}(p)\epsilon^{\nu}_{-}(p) + \epsilon^{\mu}_{-}(p)\epsilon^{\nu}_{+}(p) + 2\epsilon^{\mu}_{0}(p)\epsilon^{\nu}_{0}(p)\right) \tag{30}$$

We have to consider 10 cases:

- in-/outgoing
- polarisations ± 2, ± 1, 0

⟨ *graviton wave functions* ? ⟩ ≡
⟨ *ingoing graviton wave functions* ? ⟩
⟨ *outgoing graviton wave functions* ? ⟩◇
Macro referenced in ?.

Since we have to distinguish two Lorentz indices we use auxilliary function `fDUMMY1` to denote $\epsilon^{\mu}(k,\lambda)$ and $\epsilon^{*}_{\mu}(k,\lambda)$ in the ingoing and outgoing case.

⟨ *generic graviton wave functions* ? ⟩ ≡

```
Id once inplorentz(4, ivL4?, k1?, m?) *
        inp(field1?, k1?, +2, k2?, k3?) =
     SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
     fDUMMY1(ivL2a, k1, +1) * fDUMMY1(ivL2b, k1, +1);
     Sum ivL2a, ivL2b;
  Id once inplorentz(4, ivL4?, k1?, m?) *
        inp(field1?, k1?, +1, k2?, k3?) =
     SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
     1/Sqrt2 * (
        + fDUMMY1(ivL2a, k1, +1) * fDUMMY1(ivL2b, k1,  0)
        + fDUMMY1(ivL2a, k1,  0) * fDUMMY1(ivL2b, k1, +1)
     );
     Sum ivL2, ivL1;
  Id once inplorentz(4, ivL4?, k1?, m?) *
        inp(field1?, k1?,  0, k2?, k3?) =
     SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
     1/Sqrt2/Sqrt3 * (
        + fDUMMY1(ivL2a, k1, +1) * fDUMMY1(ivL2b, k1, -1)
        + fDUMMY1(ivL2a, k1, -1) * fDUMMY1(ivL2b, k1, +1)
        + 2 * fDUMMY1(ivL2a, k1, 0) * fDUMMY1(ivL2b, k1, 0)
     );
     Sum ivL2, ivL1;
  Id once inplorentz(4, ivL4?, k1?, m?) *
        inp(field1?, k1?, -1, k2?, k3?) =
     SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
     1/Sqrt2 * (
        + fDUMMY1(ivL2a, k1, -1) * fDUMMY1(ivL2b, k1,  0)
        + fDUMMY1(ivL2a, k1,  0) * fDUMMY1(ivL2b, k1, -1)
     );
```

```
   Sum ivL2, ivL1;
Id once inplorentz(4, ivL4?, k1?, m?) *
      inp(field1?, k1?, -2, k2?, k3?) =
   SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
   fDUMMY1(ivL2a, k1, -1) * fDUMMY1(ivL2b, k1, -1);
   Sum ivL2a, ivL2b;
◇
```

Macro referenced in ?, ?.

⟨ *ingoing graviton wave functions* ? ⟩ ≡
⟨ *generic graviton wave functions* ? ⟩
Id fDUMMY1(ivL2?, k1?, +1) =
⟨ *expression for massive* $\varepsilon(k, +1)$? ⟩;
Id fDUMMY1(ivL2?, k1?, 0) =
⟨ *expression for massive* $\varepsilon(k, 0)$? ⟩;
Id fDUMMY1(ivL2?, k1?, -1) =
⟨ *expression for massive* $\varepsilon(k, -1)$? ⟩;◇
Macro referenced in ?.

⟨ *outgoing graviton wave functions* ? ⟩ ≡
⟨ *generic graviton wave functions* ? ⟩
Id fDUMMY1(ivL2?, k1?, +1) =
⟨ *expression for massive* $\varepsilon^*(k, +1)$? ⟩;
Id fDUMMY1(ivL2?, k1?, 0) =
⟨ *expression for massive* $\varepsilon^*(k, 0)$? ⟩;
Id fDUMMY1(ivL2?, k1?, -1) =
⟨ *expression for massive* $\varepsilon^*(k, -1)$? ⟩;◇
Macro referenced in ?.

7. The Colour Part of the propagators

This section is not at the main theme of this document but for historical reasons these replacement rules are expected in the file `propagators.hh`. The colour part of a propagator for all non-trivial representations is replaced by a Kronecker-δ. The trivial representation is just ignored.

⟨ *colour part of the propagators* ? ⟩ ≡

```
Id propcolor( 3, iv1?, iv2?) = d_(iv1, iv2);
  Id propcolor(-3, iv1?, iv2?) = d_(iv1, iv2);
  Id propcolor( 8, iv1?, iv2?) = d_(iv1, iv2);
  Id propcolor( 1, iv1?, iv2?) = 1;◇
```

Macro referenced in ?.

Appendix A. Index of Symbols

Appendix B. Index of Macros

Appendix C. Index of Files

References

[1] M. Böhm, A. Denner, H. Joos: *GAUGE THEORIES of the Strong and Electroweak Interaction*, B. G. Teubner, Stuttgart, Leipzig, Wiesbaden, 3rd edition, 2001.

[2] Z. Xu, D. H. Zhang and L. Chang, *Helicity Amplitudes for Multiple Bremsstrahlung in Massless Nonabelian Gauge Theories*, Nucl. Phys. B **291** (1987) 392.

[3] W. Kilian, T. Ohl and J. Reuter, *WHIZARD: Simulating Multi-Particle Processes at LHC and ILC*, arXiv:0708.4233 [hep-ph].

E-mail address: thomasr@nikhef.nl
URL: http://www.nikhef.nl/~thomasr